>

Partial Derivatives

```
> restart:with(plots):with(plottools):
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,
  70]);
```

Visualizing Partial Derivatives on a Graph

We begin by entering the function z = $f(x, y) = x^2 - z^2$ and plotting its graph.

```
> z:=x^2-y^2;
```

$$z := x^2 - y^2$$

```
> p0:=plot3d(z,x=-6..6,y=-6..6,style=patchnogrid):
  display(p0);
```



We use the unapply command to rewrite the Maple expression **z** as the Maple function **Z**. We will use both forms in the worksheet.

```
> Z:=unapply(z,x,y);
```

$$Z := (x, y) \rightarrow x^2 - y^2$$

We focus on the point (5, 3, 16) on the graph of z.

```
> p0:=plot3d(z,x=-6..6,y=-6..6,style=patchnogrid):
  p1:=sphere([5,3,16],.2):
  display(p0,p1);
```



If you look closely, you can find the point in the plot above. To visualize the partial derivative with respect to x at the point (5, 3, 16), we view the intersection of the graph with the plane $y = 3$.

```
> p0:=plot3d(z,x=-6..6,y=-6..6,style=patchnogrid):
  p1:=sphere([5,3,16],.4):
  p2:=plot3d([t,3,t^2-9],t=-6..6,u=0..2,thickness=3):
  p3:=plot3d([t,3,u],t=-6..6,u=-30..30,color=cyan,style=wireframe)
  :
  display(p0,p1,p2,p3);
```



Staying within the plane $y = 3$, we plot the tangent line to the curve at the point (5, 3, 16).

```
> p0:=plot3d(z,x=-6..6,y=-6..6,transparency=.6):
  p1:=sphere([5,3,16],.4):
  p2:=plot3d([t,3,t^2-9],t=-6..6,u=0..2,thickness=3):
  p3:=plot3d([5+t,3,16+10*t],t=-11..1,u=0..2,thickness=3):
  display(p0,p1,p2,p3,view=-40..40);
```



We view the curve and tangent line as projected onto the x-z plane.

```
> display(p1,p2,p3,view=-40..40,orientation=[90,90]);
```



The slope of this line in the x-z plane is the partial derivative of z = f(x, y) with respect to x at the point (5, 3, 16). We use Maple to first indicate this partial derivative using Diff, then find it for any value of x using diff, and finally evaluate it at $x = 5$.

```
> fx:=Diff(z,x);
> fx:=diff(z,x);
> "fx(5,3,16)"=eval(fx,x=5);
```

$$fx := \frac{\partial}{\partial x}\left(x^2 - y^2\right)$$

$$fx := 2\,x$$

$$\text{"fx(5,3,16)"} = 10$$

When using the Maple function notation, we use the differential operator D to find partial derivative. The **1** following the **D** indicates that we are taking the derivative with respect to the first variable, **x**.

```
> Fx:=D[1](Z);
  "Fx(5,3)" = Fx(5,3);
```

$$Fx := (x, y) \rightarrow 2\,x$$

$$\text{"Fx(5,3)"} = 10$$

Next, to visualize the partial derivative with respect to y at the point (5, 3, 16), we view the intersection of the graph with the plane $x = 5$.

```
> p0:=plot3d(z,x=-6..6,y=-6..6,style=patchnogrid):
```

```
p1:=sphere([5,3,16],.4):
p2:=plot3d([5,t,25-t^2],t=-6..6,u=0..2,thickness=3):
p3:=plot3d([5,t,u],t=-6..6,u=-30..30,color=cyan,style=wireframe)
:
display(p0,p1,p2,p3);
```



Staying within the plane $x = 5$, we plot the tangent line to the curve at the point (5, 3, 16).

```
> p0:=plot3d(z,x=-6..6,y=-6..6,transparency=.6):
  p1:=sphere([5,3,16],.4):
  p2:=plot3d([5,t,25-t^2],t=-6..6,u=0..2,thickness=3):
  p3:=plot3d([5,3+t,16-6*t],t=-9..3,u=0..2,thickness=3):
  display(p0,p1,p2,p3,view=-40..40);
```



We view the curve and tangent line as projected onto the y-z plane.

```
> display(p1,p2,p3,view=-40..40,orientation=[0,90],
  labeldirections=[vertical,vertical,vertical]);
```



The slope of this line in the y-z plane is the partial derivative of z = f(x, y) with respect to y at the point (5, 3, 16). We use Maple to first indicate this partial derivative using Diff, then find it for any value of y using diff, and finally evaluate it at $y = 3$.

```
> fy:=Diff(z,y);
> fy:=diff(z,y);
> "fy(5,3,16)"=eval(fy,y=3);
```

$$fy := \frac{\partial}{\partial y}\left(x^2 - y^2\right)$$

$$fy := -2\,y$$

$$\text{"fy(5,3,16)"} = -6$$

When using the Maple function notation, we use the differential operator D to find partial derivatives. The **2** following the **D** indicates that we are taking the derivative with respect to the first variable, **y**.

```
> Fy:=D[2](Z);
  "Fy(5,3)" = Fy(5,3);
```

$$Fy := (x, y) \rightarrow -2\,y$$

$$\text{"Fy(5,3)"} = -6$$

```
>
```

Estimating Partial Derivatives from a Contour Diagram

We look in the first quadrant at the level curves for our function corresponding to levels of z equalling multiples of 4 from -40 to 40. Beginning at the point (x, y) = (5, 3), we look both at a horizontal movement along the x-axis from (5, 3) to (6, 3) and a vertical movement along the y-axis from (5, 3) to (5, 5).

```
> p0:=contourplot(z,x=-8..8,y=-8..8,contours=[seq(4*i,i=-10..10)],
  scaling=constrained):
  p1:=plot([t,3,t=5..6],color=blue,thickness=3):
  p2:=plot([5,t,t=3..5],color=green,thickness=3):
  T:=textplot({[4.5,2.8,"(5, 3)"],[6.5,2.8,"(6, 3)"],[4.5,5.2,"(5,
  5)"]},align={above,right},font=[TIMES,BOLD,14]):
  display(p0,p1,p2,T,view=[0..8,0..8]);
```



Note that the yellow contours correspond to positive levels of z, while the red contours correspond to negative levels of z. Reading the diagram, the point (5, 3) lies at level z = 16, the point (5, 5) lies at level z = 0, and we estimate that the point (6, 3) lies at level z = 27.

We estimate the partial derivative of f with respect to x by taking the difference quotient along the horizontal blue segment.

```
> fx:=(27-16)/(6-5);
```

$$fx := 11$$

This estimate compares favorably with the exact partial derivative fx = 10 computed above. Next we estimate the partial derivative of f with respect to y by taking the difference quotient along the vertical green segment.

```
> fy:=(0-16)/(5-3);
```

$$fy := -8$$

This doesn't compare quite as favorably with the with the exact partial derivative fy = -6 computed above, but we are also estimating over a larger interval.

Estimating Partial Derivatives from a Table of Values

We consider the partial table of values in the spreadsheet below.

```
>
```

SpreadSheet002

	A	B	C	D	E	F	G
1		*x*	3	4	5	6	7
2	*y*						
3	1		8	15	24	35	48
4	2		5	12	21	32	45
5	3		0	7	16	27	40
6	4			0	9	20	33
7	5		-16	-9	0	11	24
8							

```
>
```

We will use an averaging technique learned in Calculus I to estimate fx(5, 3), the point indicated in red. We will take the difference quotient using the points with magenta coloring to the left and right of (5, 3).

```
> fx:=(27-7)/(6-4);
```

$$fx := 10$$

This estimate is the exact value, the average of estimating from the right and the left. We can't always expect the exact value, but averaging usually gives better approximations that one-sided ones, 9 from the left and 11 from the right here.

We will the averaging technique to estimate fy(5, 3), the point indicated in red. We will take the difference quotient using the points with green coloring above and below (5, 3)

```
> fy:=(9-21)/(6-4);
```

$$fy := -6$$

Again, we happen to get the exact answer. The one-sided approximations in this case would be -5 from above and -7 from below.

Second Order Partial Derivatives

Let's look at another function.

```
> z:=x*exp(-x^2-y^2);
```

$$z := x\,\mathrm{e}^{-x^2 - y^2}$$

```
> Fx:=diff(z,x);
```

$$Fx := \mathrm{e}^{-x^2 - y^2} - 2\,x^2\,\mathrm{e}^{-x^2 - y^2}$$

```
> Fy:=diff(z,y);
```

$$Fy := -2\,x\,y\,\mathrm{e}^{-x^2 - y^2}$$

We do the second order partials by using Diff.

```
> Fxx:=Diff(Diff(z,x),x);
```

$$Fxx := \frac{\partial^2}{\partial x^2}\left(x\,\mathrm{e}^{-x^2 - y^2}\right)$$

```
> Fxy:=Diff(Diff(z,x),y);
```

$$Fxy := \frac{\partial^2}{\partial y\,\partial x}\left(x\,\mathrm{e}^{-x^2 - y^2}\right)$$

```
> Fyx:=Diff(Diff(z,y),x);
```

$$Fyx := \frac{\partial^2}{\partial x\,\partial y}\left(x\,\mathrm{e}^{-x^2 - y^2}\right)$$

```
> Fyy:=Diff(Diff(z,y),y);
```

$$Fyy := \frac{\partial^2}{\partial y^2}\left(x\,\mathrm{e}^{-x^2 - y^2}\right)$$

Of course, one can evaluate any of these by using value.

```
> Fyy:=value(Diff(Diff(z,y),y));
```

$$Fyy := -2\,x\,\mathrm{e}^{-x^2 - y^2} + 4\,x\,y^2\,\mathrm{e}^{-x^2 - y^2}$$

Now let's go straight to the values of the derivatives.

```
> Fxx:=diff(Fx,x);
```

$$Fxx := -6\,x\,\mathrm{e}^{-x^2 - y^2} + 4\,x^3\,\mathrm{e}^{-x^2 - y^2}$$

```
> Fxy:=diff(Fx,y);
```

$$Fxy := -2\,y\,\mathrm{e}^{-x^2 - y^2} + 4\,x^2\,y\,\mathrm{e}^{-x^2 - y^2}$$

```
> Fyx:=diff(Fy,x);
```

$$Fyx := -2\,y\,\mathrm{e}^{-x^2 - y^2} + 4\,x^2\,y\,\mathrm{e}^{-x^2 - y^2}$$

Notice that the mixed partials are equal. Just for variety,

```
> Fyy:=diff(diff(z,y),y);
```

$$Fyy := -2\,x\,\mathrm{e}^{-x^2 - y^2} + 4\,x\,y^2\,\mathrm{e}^{-x^2 - y^2}$$

Now suppose the function is given as a Maple function.

```
> F:=(x,y)->x*exp(-x^2-y^2);
```

$$F := (x, y) \rightarrow x\,\mathrm{e}^{-x^2 - y^2}$$

In this situation, we could proceed exactly as above, just replacing each occurance of z with $F(x, y)$.
But another option is to use the differential operator D.

```
> Fx:=D[1](F);
```

$$Fx := (x, y) \rightarrow \mathrm{e}^{-x^2 - y^2} - 2\,x^2\,\mathrm{e}^{-x^2 - y^2}$$

```
> Fy:=D[2](F);
```

$$Fy := (x, y) \to -2\,x\,y\,e^{-x^2 - y^2}$$

```
> Fxx:=D[1,1](F);
```

$$Fxx := (x, y) \to -6\,x\,e^{-x^2 - y^2} + 4\,x^3\,e^{-x^2 - y^2}$$

```
> Fxy:=D[1,2](F);
```

$$Fxy := (x, y) \to -2\,y\,e^{-x^2 - y^2} + 4\,x^2\,y\,e^{-x^2 - y^2}$$

```
> Fxy:=D[2](D[1](F));
```

$$Fxy := (x, y) \to -2\,y\,e^{-x^2 - y^2} + 4\,x^2\,y\,e^{-x^2 - y^2}$$

```
> Fyx:=D[2,1](F);
```

$$Fyx := (x, y) \to -2\,y\,e^{-x^2 - y^2} + 4\,x^2\,y\,e^{-x^2 - y^2}$$

```
> Fyy:=D[2,2](F);
```

$$Fyy := (x, y) \to -2\,x\,e^{-x^2 - y^2} + 4\,x\,y^2\,e^{-x^2 - y^2}$$

```
>
```